The Computation of Elementary Unitary Matrices

R.B. Lehoucq�

April 20, 1994

Abstract

The construction of elementary unitary matrices that transform a complex vector to a real multiple of e_1, the first column of the identity matrix, are studied. We survey the two well known forms and present what appears to be relatively unknown third form implemented by LAPACK subroutine CLARFG.

1 Introduction

Define the elementary Hermitian (or unitary) matrix U = I − 2ww^H, (1) where w^Hw = 1. Easily verified is that U is both Hermitian and unitary and is also refered to as a Householder matrix. The matrix U as defined by ( 1) is a special case of the more general class of elementary matrices defined by E(u, v; σ) = I − σuv^H. (2) See [2], [3] and [4] for further details. The latter reference is a comprehensive study of elementary matrices, their properties and extensions to block implementations.

In [3] Wilkinson states that it is not possible, in general, to find an elementary matrix U such that

Ux = αe_1, (3)

for a complex vector x with α a real number. Note that U is unitary implies that |α| = ‖x‖_2 . If the above relation were satisfied then x^Hαe_1 = x^H Ux. The righthand side is a real number since U is Hermitian. But if e^T_1 x has a non-zero imaginary part we have a contradiction. The necessary condition is that x^H e_1 = e^T_1 x. Equivalently e^T_1 x = ξ_1 must have zero imaginary part.

The purpose of this note is to review and examine the case of constructing elementary unitary matrices that satisfy (3) for a complex vector x. The next section

�Department of Computational and Applied Mathematics, Rice University(lehoucq@rice.edu). This work was supported by DARPA under contract TV-ORA4466.01

discusses the approach suggested by Wilkinson. The third section introduces yet another approach due to Hammarling and Du Croz. It has been implemented in the NAG Fortran Library subroutine F06HRF, see [1]. A slightly different implementation is the LAPACK subroutine CLARFG. The details of the software implementation for CLARFG are in section four. In fact what led to this study is attempting to understand the differences between the Wilkinson approach and the alternate formulation implemented by LAPACK, [6]. Section five generalizes the Hammarling–Du Croz form for transforming a complex vector x into another complex vector y. The last section contains the source code for subroutine CLARFG.

2 The Wilkinson Approach

Wilkinson suggests the following modification in Chapter 2, section 45 of [3]. If e^T_1 x = ξ_1 has non-zero imaginary part then

x = e^{iθ1} [|ξ_1|, e�^{iθ1} ξ_2, . . ., e�^{iθ1} ξ_n]^T , = e^{iθ1} y, where e^{iθ1} |ξ_1| is the polar form for ξ_1. Since e^T_1 y is a real number, a Householder matrix P may be constructed satisfying (3). Set U = e�^{iθ1} P and Ux = e�^{iθ1} P x, = e�iθ1 P eiθ1 y, = P y, = αe_1.

The matrix U is not Hermitian but is unitary which is the crucial property. The value of α is not affected; |α| = ‖Ux‖_2, = |e�^{iθ1} | ‖P x‖_2, = ‖x‖_2. The matrix U constructed is not an elementary matrix. If ξ_1 is real then θ_1 is arbitrary and may be chosen equal to zero. The resulting elementary matrix is then also Hermitian.

The Wilkinson approach is used by LINPACK when working with complex data.

3 An Alternate Approach

The final form for an elementary unitary matrix studied is due to Hammarling and Du Croz, see [1] (Introduction – F06). We now derive this alternate form. Consider the general form ( 2) for an elementary unitary matrix U = E(u, v; σ). The matrix must be unitary from which it follows that I = U^H U, = (I − σuv^H)^H(I − σuv^H), = I − σ̄vu^H − σuv^H + σσ̄(u^H u)vv^H, and cancelling terms results in σσ̄(u^H u)vv^H = σ̄vu^H + σuv^H. (4) Rearranging terms gives (σσ̄(u^Hu)v − σu)v^H = σ̄vu^H, and a row space argument implies that u and v are linearly dependent. Substituting u = v into (4) results in |σ|^2‖v‖^2 = σ + σ̄, (5) = 2Re(σ), determining the required relationship between σ and v. Choosing v = x − αe_1 we have Ux = (I − σvv^H)x, = x − (σv^Hx)v, = αe_1, if σ�^1 = v^Hx. This choice of σ will satisfy (5) as we now demonstrate. First v^Hx = (x^H − αe^T_1 )x, = x^Hx − αξ_1, = α(α − ξ_1), which determines σ and ‖v‖^2 = v^Hv, = (x^H − αe^T_1 )(x − αe_1), = 2α(α − Re(ξ_1)), follows. Finally

(v^Hx)(x^Hv)(σ + σ̄) = (v^Hx)(x^Hv)( 1 v^Hx + 1 x^H v ), = x^H v + v^H x, = α(α − ξ_1) + α(α − ξ̄_1), = 2α(α − Re(ξ_1)), shows σ + σ̄ |σ|^2 = ‖v‖^2,

as claimed. The form as outlined in this section does not appear to be as widely known as the Wilkinson one. It has the benefit of transforming a complex vector x directly to real multiple of e_1. When ξ_1 is purely real then U is Hermitian. Unlike the Wilkinson variant the Hammarling–Du Croz approach results in an elementary unitary matrix.

4 Software Implementation of CLARFG

The actual implementation of the Hammarling–Du Croz variant has some slight modifications. The resulting code is an excellent example of the art of developing software from a numerical algorithm. Subroutine CLARFG determines an elementary unitary matrix U = I − τuu^H such that U^H x = αe_1 where |α| = ‖x‖_2 for α a real number. Simplifying the implementation of other algorithms in LAPACK requiring the use of elementary unitary matrices the normalization u^H e_1 = 1 is taken. This normalization is done for storage considerations and is discussed in [5]. From the previous section it follows that

U^H = I − 1 α(α − ξ̄_1) vv^H, = I − (ξ_1 − α)(ξ̄_1 − α) α(α − ξ̄_1) uu^H , = I − α − ξ_1 α uu^H , = I − τuu^H ,

where u = v ξ_1 − α , = x − αe_1 ξ_1 − α , and τ = α − ξ_1 α . Storage of U for use in further computation only requires storage for the complex τ and x may be overwritten with both α and the essential part of u, i.e. x ← [α, ξ_2 ξ_1 − α , . . ., ξ_n ξ_1 − α

]^T .

In order that the value for τ and the scaling factor ξ_1 − α have small relative error α ← −sign(Re(ξ_1))‖x‖_2, is chosen. This is a standard modification and is mentioned for the sake of completeness. The resulting τ satisfies the two properties

1 ≤ Re(τ) ≤ 2, |τ − 1| ≤ 1, when x ≠ γe_1 with γ real. If x is a real multiple of e_1 then set τ ← 0, U ← I.

The reviewer of CLARFG will notice one final point which needs explanation. The careful programmer took care not to reciprocate the number ‖x‖_2 that may fall below a certain machine dependent tolerance, SAFMIN. The value SAFMIN, computed by the LAPACK auxiliary subroutine SLAMCH is a machine dependent lower bound for numbers that may be safely reciprocated and not cause an overflow condition. If ‖x‖_2 is less than the lower bound then the vector x is scaled by a multiple of the reciprocal of SAFMIN until ‖θx‖2 is at least as large as SAFMIN. Defining the integer k to represent the number of scalings required results in θ = k 1 SAFMIN . The number τ may now be safely computed as τ ← ‖θx‖_2 − θξ_1 ‖θx‖_2 . In a similar fashion

α ← −sign(Re(ξ_1)) 1 θ (‖θx‖_2), and the essential part of u u ← 1 θξ_1 − θα [θξ_2, . . ., θξ_n]^T , are computed.

5 Generalizations

We conclude with the problem of determining an elementary unitary matrix U that satisfies Ux = (I − σvv^H)x, (6) = y, where ‖x‖_2 = ‖y‖_2. The derivation of section three requires little modification. The value of σ�^1 = v^Hx and v = x − y, v^Hx = ‖x‖^2 − y^H x, may be shown to satisfy (5) and (6).

6 Acknowledgements

The author would like to thank Jermey Du Croz for background information on the Hammarling–Du Croz approach and for encouragement. Both Jeremy Du Croz and Dan Sorensen carefully read preliminary manuscripts and their comments are appreciated. The clever scaling in section four is due to James Demmel.

References

[1] NAG Fortran Library Manual, Mark 16, Numerical Algorithms Group Ltd, Oxford, UK, 1993.

[2] A. H. Householder, The Theory of Matrices in Numerical Analysis, Blaisdell, 1964.

[3] J. H. Wilkinson, The Algebraic Eigenvalue Problem, Clarendon Press, Oxford, UK, 1965.

[4] A. A. Dubrulle, Work Notes on Elementary Matrices, Hewlett-Packard Laboratories, 1993, Technical Report HPL-93-69.

[5] G. H. Golub, C. F. Van Loan, Matrix Computations, Johns Hopkins, second edition, 1989.

[6] E. Anderson, Z. Bai, C. Bischoff, J. Demmel, J. Dongarra, J. Du Croz, A. Greenbaum, S. Hammarling, A. McKenney, S. Ostrouchov, and D. C. Sorensen, LAPACK Users’ Guide, SIAM, 1992.

[7] J. J. Dongarra, C. B. Moler, J. R. Bunch, G. W. Stewart, LINPACK Users’ Guide, SIAM, 1979.

7 LAPACK Subroutine CLARFG

      SUBROUTINE CLARFG( N, ALPHA, X, INCX, TAU )
*
* -- LAPACK auxiliary routine (version 1.1) --
* Univ. of Tennessee, Univ. of California Berkeley, NAG Ltd.,
* Courant Institute, Argonne National Lab, and Rice University
* February 29, 1992
*
* .. Scalar Arguments ..
      INTEGER INCX, N
      COMPLEX ALPHA, TAU
* ..
* .. Array Arguments ..
      COMPLEX X( * )
* ..
*
* Purpose
* =======
*
* CLARFG generates a complex elementary reflector H of order n, such
* that
*
* H’ * ( alpha ) = ( beta ), H’ * H = I.
* ( x ) ( 0 )
*

* where alpha and beta are scalars, with beta real, and x is an
* (n-1)-element complex vector. H is represented in the form
*
* H = I - tau * ( 1 ) * ( 1 v’ ) ,
* ( v )
*
* where tau is a complex scalar and v is a complex (n-1)-element
* vector. Note that H is not hermitian.
*
* If the elements of x are all zero and alpha is real, then tau = 0
* and H is taken to be the unit matrix.
*
* Otherwise 1 <= real(tau) <= 2 and abs(tau-1) <= 1 .
*
* Arguments
* =========
*
* N (input) INTEGER
* The order of the elementary reflector.
*
* ALPHA (input/output) COMPLEX
* On entry, the value alpha.
* On exit, it is overwritten with the value beta.
*
* X (input/output) COMPLEX array, dimension
* (1+(N-2)*abs(INCX))
* On entry, the vector x.
* On exit, it is overwritten with the vector v.
*
* INCX (input) INTEGER
* The increment between elements of X. INCX <> 0.
*
* TAU (output) COMPLEX
* The value tau.
*
* ===================================================================
*
* .. Parameters ..
      REAL ONE, ZERO
      PARAMETER ( ONE = 1.0E+0, ZERO = 0.0E+0 )
* ..
* .. Local Scalars ..
      INTEGER J, KNT
      REAL ALPHI, ALPHR, BETA, RSAFMN, SAFMIN, XNORM
* ..
* .. External Functions ..
      REAL SCNRM2, SLAMCH, SLAPY3

      COMPLEX CLADIV
      EXTERNAL CLADIV, SCNRM2, SLAMCH, SLAPY3
* ..
* .. Intrinsic Functions ..
      INTRINSIC ABS, AIMAG, CMPLX, REAL, SIGN
* ..
* .. External Subroutines ..
      EXTERNAL CSCAL, CSSCAL
* ..
* .. Executable Statements ..
*
      IF( N.LE.0 ) THEN
         TAU = ZERO
         RETURN
      END IF
*
      XNORM = SCNRM2( N-1, X, INCX )
      ALPHR = REAL( ALPHA )
      ALPHI = AIMAG( ALPHA )
*
      IF( XNORM.EQ.ZERO .AND. ALPHI.EQ.ZERO ) THEN
*
* H = I
*
         TAU = ZERO
      ELSE
*
* general case
*
         BETA = -SIGN( SLAPY3( ALPHR, ALPHI, XNORM ), ALPHR )
         SAFMIN = SLAMCH( ’S’ )
         RSAFMN = ONE / SAFMIN
*
         IF( ABS( BETA ).LT.SAFMIN ) THEN
*
* XNORM, BETA may be inaccurate; scale X and recompute them
*
            KNT = 0
   10 CONTINUE
            KNT = KNT + 1
            CALL CSSCAL( N-1, RSAFMN, X, INCX )
            BETA = BETA*RSAFMN
            ALPHI = ALPHI*RSAFMN
            ALPHR = ALPHR*RSAFMN
            IF( ABS( BETA ).LT.SAFMIN )
     $ GO TO 10
*

* New BETA is at most 1, at least SAFMIN
*
            XNORM = SCNRM2( N-1, X, INCX )
            ALPHA = CMPLX( ALPHR, ALPHI )
            BETA = -SIGN( SLAPY3( ALPHR, ALPHI, XNORM ), ALPHR )
            TAU = CMPLX( ( BETA-ALPHR ) / BETA, -ALPHI / BETA )
            ALPHA = CLADIV( CMPLX( ONE ), ALPHA-BETA )
            CALL CSCAL( N-1, ALPHA, X, INCX )
*
* If ALPHA is subnormal, it may lose relative accuracy
*
            ALPHA = BETA
            DO 20 J = 1, KNT
               ALPHA = ALPHA*SAFMIN
   20 CONTINUE
         ELSE
            TAU = CMPLX( ( BETA-ALPHR ) / BETA, -ALPHI / BETA )
            ALPHA = CLADIV( CMPLX( ONE ), ALPHA-BETA )
            CALL CSCAL( N-1, ALPHA, X, INCX )
            ALPHA = BETA
         END IF
      END IF
*
      RETURN
*
* End of CLARFG
*
      END